
SI-1. JAGS MODEL

The JAGS model follows the paper’s specification, and is quite general. It can handle any

number of predictors, any number of dimensions, and any number of cut-offs within the

different dimensions. It assumes that there is some observation-level data that can predict

participation, and once participation is attained, it predicts the level of participation on

all d dimensions. The estimated variance-covariance matrix can be used to determine the

correlations between dimensions.

model{

#y is a matrix of outcomes of dimension NxD

#N is the number of observations

#D is the number of dimensions

#Z is the matrix of first-stage predictors of dimension NxK

#K is the number of first-stage predictors, including an intercept

#X is an array of predictors for the second stage

#n.cut is a scalar, the number of cut points

#c is a vector, the second cut point

#R is the inverse-Wishart prior matrix

for(i in 1:N){

#if outcomes have different number of levels, change n.cut to a vector

#first stage

S[i] <- max(min(S.star[i],1),0)

probit(S.star[i]) <- s.star[i] #participation regime probability

s.star[i] <- inprod(Z[i,], gamma[1:K])

#second stage

for(d in 1:D){

Xnew[d,i] <- inprod(X[i,,d], beta[1:J,d])

}

y.star[i, 1:D] ~ dmnorm(Xnew[,i], invW[,])

for(d in 1:D){

for (i.cut in 1:n.cut){

probit(Q[i,i.cut,d]) <- tau.unsorted[i.cut, d] - y.star[i,d]

}

}

for(d in 1:D){

p[i,d,1] <- (1-S[i]) + S[i]*max(min(Q[i,1,d],1),0)

for(i.cut in 2:n.cut){

p[i,d,i.cut] <- S[i]*(max(min(Q[i,i.cut,d],1),0) - max(min(Q[i,i.cut-1,d],1),0))

}

p[i,d,n.cut+1] <- S[i]*(1 - max(min(Q[i,2,d],1),0))

y[i,d] ~ dcat(p[i,d,])

}

}

for(j in 1:J){

for(d in 1:D){

beta[j,d] ~ dnorm(0,0.0001)

}

}

for(k in 1:K){

gamma[k] ~ dnorm(0,0.0001)

}

for(d in 1:D){

tau.unsorted[1,d] <- 0

tau.unsorted[2,d] <- c[d]

#uncomment below for more than three outcomes

#for(i.cut in 3:n.cut){

# tau.unsorted[i.cut,d] <- tau[i.cut, d]

# tau[i.cut, d] ~ dlnorm(0,0.025)

#}

}

#can change D+1 to be anything greater than or equal to D

#can pass this in as a parameter

invW[1:D,1:D] ~ dwish(R[,],D+1)

Sigma[1:D,1:D] <- inverse(invW[,])

for(d in 1:D){

vars[d] <- Sigma[d,d]

}

for(l in 1:(D-1)){

for(q in l:(D-1)){

rho[q+l-1] <- Sigma[l,q+1]/sqrt(vars[l]*vars[q+1])

}

}

}

SI-2. SIMULATION EXERCISE

Table SI-1 shows the true parameters for the data generating process for the first round of
simulations. Table SI-2 shows the true parameters for the data generating process for the
second round of simulations.

Table SI-1: True parameters for the first round of simulations.
Simulation γ0 γ1 β10 β11 β20 β21 β30 β31

1 −1.5 4 0 2 0 2.5 0 2
2 −1.5 5 0 2 0 2.5 0 2
3 −1.5 6 0 2 0 2.5 0 2
4 −1.5 7 0 2 0 2.5 0 2
5 −1.5 8 0 2 0 2.5 0 2
6 −1.5 9 0 2 0 2.5 0 2
7 −1.5 10 0 2 0 2.5 0 2
8 −1.5 11 0 2 0 2.5 0 2

Note: Each simulation was repeated 100 times. All
correlation coefficients were set to zero. The first-stage
parameters are in the second and third columns. The
second-stage parameters are in the latter six columns.
The subscripts on these parameters denote first the di-
mension, and second the indicator for the parameter.
Zeros refer to intercepts, and ones refer to the variable
of interest.

Table SI-2: True parameters for the second round of simulations.
Simulation γ0 γ1 β10 β11 β20 β21 β30 β31 ρ12 ρ13 ρ23

1 −1.5 9 0 2 0 2.5 0 2 −.8 0 0
2 −1.5 9 0 2 0 2.5 0 2 −.6 0 0
3 −1.5 9 0 2 0 2.5 0 2 −.4 0 0
4 −1.5 9 0 2 0 2.5 0 2 −.2 0 0
5 −1.5 9 0 2 0 2.5 0 2 0 0 0
6 −1.5 9 0 2 0 2.5 0 2 .2 0 0
7 −1.5 9 0 2 0 2.5 0 2 .4 0 0
8 −1.5 9 0 2 0 2.5 0 2 .6 0 0
9 −1.5 9 0 2 0 2.5 0 2 .8 0 0
10 −1.5 9 0 2 0 2.5 0 2 −.8 .64 −.51
11 −1.5 9 0 2 0 2.5 0 2 −.6 .36 −.22
12 −1.5 9 0 2 0 2.5 0 2 −.4 .16 −.06
13 −1.5 9 0 2 0 2.5 0 2 −.2 .04 −.01
14 −1.5 9 0 2 0 2.5 0 2 0 0 0
15 −1.5 9 0 2 0 2.5 0 2 .2 .04 .01
16 −1.5 9 0 2 0 2.5 0 2 .4 .16 .06
17 −1.5 9 0 2 0 2.5 0 2 .6 .36 .22
18 −1.5 9 0 2 0 2.5 0 2 .8 .64 .51
19 −1.5 9 0 2 0 2.5 0 2 −.8 −.8 .64
20 −1.5 9 0 2 0 2.5 0 2 −.6 −.6 .36
21 −1.5 9 0 2 0 2.5 0 2 −.4 −.4 .16
22 −1.5 9 0 2 0 2.5 0 2 −.2 −.2 .04
23 −1.5 9 0 2 0 2.5 0 2 0 0 0
24 −1.5 9 0 2 0 2.5 0 2 .2 .2 .04
25 −1.5 9 0 2 0 2.5 0 2 .4 .4 .16
26 −1.5 9 0 2 0 2.5 0 2 .6 .6 .36
27 −1.5 9 0 2 0 2.5 0 2 .8 .8 .64

Note: Each simulation was repeated 100 times. The first-stage parame-
ters are in the second and third columns. The second-stage parameters
are in the fourth through tenth columns. For the first- and second-stage
parameters, the subscripts denote first the dimension, and second the in-
dicator for the parameter. Zeros refer to intercepts, and ones refer to the
variable of interest. The final columns are the correlation coefficients of
the second stage. The subscripts denote the dimensions (e.g. a subscript
of 12 refers to the correlation between the residuals of the first and second
dimensions). The first nine set the second and third correlations, ρ13 and
ρ23, to zero, and ρ12 varies from −0.8 to 0.8 by 0.2. The second nine keep
the same ρ12 shift, setting ρ13 to ρ212 and ρ23 to ρ312. The final nine again
maintain the same ρ12 shift and set ρ13 to ρ12 and ρ23 = ρ212. This choice
stemmed partly from the need to generate positive definite matrices.

SI-3. PRESIDENTIAL CAMPAIGNS IN MEXICO

To better assess how our inferences would be changed if the main application in the paper
did not allow for correlations, I present a model that restricts the correlation parameters to
be zero. This is similar to the simulation exercises comparing ZIMVOP to an uncorrelated
multivariate probit. The first-step estimates are nearly identical between models, but the
second-step estimates vary fairly substantially, with some estimates reliable in one model
and not in the other. A main parameter of interest, Previous vote, though not reliable at
the 95% level for the PRI, is not even reliable at the 90% in the second model. The results,
presented next to each other, are shown in Figure SI-1. Not only would we be losing the
information provided by the correlation estimates, the simulations suggest that we should
trust the posteriors from ZIMVOP.

Figure SI-1: Comparing ZIMVOP to a model without correlations on the main application.

Note: The second-step results from ZIMVOP are presented on the left panel, and the results
not allowing correlations is on the right panel. As can be seen, the posteriors are fairly
different. Not only would we be losing the information provided by the correlation estimates,
the simulations suggest that we should trust the posteriors from ZIMVOP.

